# $Id$
# Maintainer: Eric Belanger <eric@archlinux.org>
# Contributor: Jan de Groot <jgc@archlinux.org>

pkgname=font-bitstream-speedo
pkgver=1.0.2
pkgrel=4
pkgdesc="Bitstream Speedo fonts"
arch=('any')
url="https://xorg.freedesktop.org/"
license=('custom')
depends=('xorg-mkfontdir' 'xorg-mkfontscale' 'fontconfig')
source=(https://xorg.freedesktop.org/releases/individual/font/${pkgname}-${pkgver}.tar.bz2)
sha256sums=('0708fe0046947b88c082b55844af8a861741dcdf087f0624bdf23fb773d76763')

build() {
  cd "${srcdir}/${pkgname}-${pkgver}"
  ./configure --prefix=/usr \
      --with-fontdir=/usr/share/fonts/speedo 
  make
}

package() {
  cd "${srcdir}/${pkgname}-${pkgver}"
  make DESTDIR="${pkgdir}" install
  rm -f "${pkgdir}/usr/share/fonts/speedo"/fonts.*
  
  install -Dm644 COPYING "${pkgdir}/usr/share/licenses/${pkgname}/COPYING"
}
